
	Reason for change:
	Further to change 4 of R5s220577, there are some further modifications required to the logic of the function f_NR5GC_IRAT_S1ToN1_SessionModification() to handle the possible PDU session modification procedures following S1 to N1 transfer (with multiple active PDN connections). In the current implementation, if a PDU SESSION MODIFICATION COMMAND has be`en sent in response to a PDU SESSION MODIFICATION REQUEST, and while waiting for the PDU SESSION MODIFICATION COMPLETE the function receives a 2nd PDU SESSION MODIFICATION REQUEST on SRB2, this will be stored in the variable v_PduSessionModificationRequestList[L]. On receipt of the pending PDU SESSION MODIFICATION COMPLETE, the counter K is incremented and the variable v_RxdMsg is reset to false. Because the pending PDU SESSION MODIFICATION REQUEST is being held in the local variable v_PduSessionModificationRequestList and not the external default handler, there is currently no trigger to process this – the check “if (not (v_WaitForComplete) and f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L)” will return false and the execution will re-enter the “alt” statement in the “else” branch to wait for a further message.

Note : This behaviour happens when the 2nd PDU SESSION MODIFICATION REQUEST is received on SRB2 and so matches with the “SRB.recieve” statement in the alt block (which will have v_SRB set to SRB2). This happens in the connected mode handover cases where SRB2 is already established. In the idle mode cases where SRB2 is not yet established, the 2nd PDU SESSION MODIFICATION REQUEST will be received on SRB1 and so will not match with the “SRB.recieve” statement. In this case it is caught and stored by the external default handler and is picked up by the check  f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L after PDU SESSION MODIFICATION COMPLETE is processed. For this reason the idle mode cases are working correctly with the current implementation but there is an issue with the handover test cases e.g. 8.1.4.2.2.1.

	
	

	Summary of change:
	After receiving a message (exiting the “alt” block with v_RxdMsg=true), move the checking for PDU SESSION MODIFICATION COMPLETE to before PDU SESSION MODIFICATION REQUEST (in a separate “if” block). On completing processing of PDU SESSION MODIFICATION COMPLETE, check if there is any pending PDU SESSION MODIFICATION REQUEST in v_PduSessionModificationRequestList[] variable, and if so load this into the variable v_GSMMsg so that it will be processed next.

Note : The fatal error in the “else” block is now removed so that this is not triggered in the case where there is no pending PDU SESSION MODIFICATION REQUEST following processing of PDU SESSION MODIFICATION COMPLETE.

1. [bookmark: _Toc122434485][bookmark: _Toc126242285]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D22wk49 related to the title of this CR.
Contact:	Shaun Harry 
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc126242286]Corrections required
1.1 [bookmark: _Toc122605392][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc126242287]Change 1
	Function name
	f_NR5GC_IRAT_S1ToN1_SessionModification()

	Reason for change
	Further to change 4 of R5s220577, there are some further modifications required to the logic of this function to handle the required PDU session modification procedures following S1 to N1 transfer (with multiple active PDN connections). In the current implementation, if a PDU SESSION MODIFICATION COMMAND has been sent in response to a PDU SESSION MODIFICATION REQUEST, and while waiting for the PDU SESSION MODIFICATION COMPLETE the function receives a 2nd PDU SESSION MODIFICATION REQUEST on SRB2, this will be stored in the variable v_PduSessionModificationRequestList[L]. On receipt of the pending PDU SESSION MODIFICATION COMPLETE, the counter K is incremented and the variable v_RxdMsg is reset to false. Because the pending PDU SESSION MODIFICATION REQUEST is being held in the local variable v_PduSessionModificationRequestList and not the external default handler, there is currently no trigger to process this – the check “if (not (v_WaitForComplete) and f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L)” will return false and the execution will re-enter the “alt” statement in the “else” branch to wait for a further message.

Note : This behaviour happens when the 2nd PDU SESSION MODIFICATION REQUEST is received on SRB2 and so matches with the “SRB.recieve” statement in the alt block (which will have v_SRB set to SRB2). This happens in the connected mode handover cases where SRB2 is already established. In the idle mode cases where SRB2 is not yet established, the 2nd PDU SESSION MODIFICATION REQUEST will be received on SRB1 and so will not match with the “SRB.recieve” statement. In this case it is caught and stored by the external default handler and is picked up by the check  f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L after PDU SESSION MODIFICATION COMPLETE is processed. For this reason the idle mode cases are working correctly with the current implementation but there is an issue with the handover test cases e.g. 8.1.4.2.2.1.

	Summary of change
	After receiving a message (exiting the “alt” block with v_RxdMsg=true), move the checking for PDU SESSION MODIFICATION COMPLETE to before PDU SESSION MODIFICATION REQUEST (in a separate “if” block). On completing processing of PDU SESSION MODIFICATION COMPLETE, check if there is any pending PDU SESSION MODIFICATION REQUEST in v_PduSessionModificationRequestList[] variable, and if so load this into the variable v_GSMMsg so that it will be processed next.

Note : The fatal error in the “else” block is now removed so that this is not triggered in the case where there is no pending PDU SESSION MODIFICATION REQUEST following processing of PDU SESSION MODIFICATION COMPLETE.

	TTCN module
	NR5GC_IRAT_CommonFunctions.ttcn

	MCC160 Comment
	



Before Change:
	  
  function f_NR5GC_IRAT_S1ToN1_SessionModification(NR_CellId_Type p_NR_CellId,
                                                   EUTRA_NR_PduSessionInfoList_Type p_PduSessionsFromLTE,
                                                   boolean p_FailWhenK_LessN := true,
                                                   NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                                                   integer p_ExpectedNumberOfNewPDUSessions := (pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection +1),  // @sic R5s220577 sic@
                                                   boolean p_DRBsAlreadyConfigured := false) runs on NR5GC_PTC return boolean
  {
    var template (value) PDUSessionInfoList_Type v_FromLTEList;
    var NG_NAS_UL_PduList_Type v_PduSessionModificationRequestList := {};
    var IntegerList_Type v_PDU_SessionIdList := {};
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_NG_NAS_UL_Pdu;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var integer v_PDU_SessionId;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)
    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;
    var boolean v_ReregisterIMS := false;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var integer N := p_ExpectedNumberOfNewPDUSessions;  // number of PDU SESSION MODIFICATION REQUEST messages being expected
    var integer K := 0;   // number of PDU SESSION MODIFICATION REQUEST messages already processed (i.e. the corresponding PDU SESSION MODIFICATION COMMAND and PDU SESSION MODIFICATION COMPLETE have been sent/received)
    var integer L := 0;   // number of PDU SESSION MODIFICATION REQUEST messages being received
    var boolean v_WaitForComplete := false;
    var template (present) NG_UE_SM_Cap v_NG_UE_SM_Cap := ?;                         // according to 38.508-1 Table 4.5A.2C.2.3-1
    var template (present) IntegrityProtMaxDataRate v_IntegrityProtMaxDataRate := ?; // according to 38.508-1 Table 4.5A.2C.2.3-1
    var integer i;
    var SRBs_AlreadyConfigured v_SRBs := SRB1;
    var NR_RadioBearerId_Type v_SRB := p_SRB;  // @sic R5s210672 sic@
    var IntegerList_Type v_TransferredSessions;
    var PDU_PDN_DNN_Type v_PDU_Type := Internet_DNN; // assume this for now @sic R5-206296 sic@
    var boolean v_RxdMsg := false;
    var boolean v_WaitForRRCComplete := false;
    
    timer t_Wait := 10.0; // @sic R5-204330 sic@

    // Build the GSMMobilityInfo for all received PDUs, but don't store it (some may be established, rather than modified)
    if (ispresent(v_GMM_MobilityInfo.SessionStatus)) { // @sic R5-206290 sic@
      v_TransferredSessions := f_NG_ConvertPDU_SessionStatus(valueof(v_GMM_MobilityInfo.SessionStatus));
      for (i := 0; i < lengthof (p_PduSessionsFromLTE); i := i + 1) {
        if (f_IntegerList_Search(v_TransferredSessions, p_PduSessionsFromLTE[i].PDU_SessionId) != -1) {
          // don't store if not in transferred list
          if (p_PduSessionsFromLTE[i].IsIMS) {  // @sic R5-206296 sic@
            v_PDU_Type := IMS_DNN;
          }
          v_FromLTEList[i] := cs_GSM_MobilityInfo(int2oct(p_PduSessionsFromLTE[i].PDU_SessionId, 1), -, v_PDU_Type); // @sic R5s200564 sic@
        }
      }
    }
        
    t_Wait.start;  // @sic R5-204330 sic@
    while ((K < N) or v_WaitForRRCComplete) { // @sic R5-215678 sic@
      if (not (v_WaitForComplete) and f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L) { // @sic R5s210520, R5s220174, R5s220577 sic@
        // 1st msg was received before entering function
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(L+1); // @sic R5s220174, R5s220577  sic@
        v_RxdMsg := true;
        t_Wait.stop;
      } else {
        alt { // @sic R5-204330 sic@
          [v_WaitForRRCComplete] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB1,
                                      cr_38508_RRCReconfigurationComplete))     // @sic R5s211280 sic@
          {
            v_WaitForRRCComplete := false;
          }
          [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp // @sic R5s201413, R5s210672 sic@
          {
            v_RxdMsg := true;
            t_Wait.stop;
          }
          [] t_Wait.timeout {
            if (p_FailWhenK_LessN and (K < N)) {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not enough PDU Session Modification Request messages");
            } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop  @sic R5-204330 sic@
              break;
            }
          }
        }
      }
      if (v_RxdMsg) { // @sic R5s210520 sic@
        v_NG_NAS_UL_Pdu := v_ReceivedAsp.Signalling.Nas[0].Pdu;
        v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;

        if (ischosen(v_GSMMsg.pdu_Session_Modification_Request)) {
          v_PduSessionModificationRequestList[L] := v_NG_NAS_UL_Pdu;
          L := L + 1;
          if (L > N) { // @sic R5-204330 sic@
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "L > N");
          }
          if (not v_WaitForComplete and (K < L)) {   // => process the next PDU SESSION MODIFICATION REQUEST
            v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];

            if (not f_Check_NG_PDUSessionModificationReq(v_GSM_MobilityInfo, valueof(v_FromLTEList), v_NG_NAS_UL_Pdu, -, -, v_NG_UE_SM_Cap, -, -, -, v_IntegrityProtMaxDataRate, -, -, -, -, -, cr_NG_Request_Type('101'B))) {     // v_GSM_MobilityInfo is out parameter of f_Check_NG_PDUSessionModificationReq
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION REQUEST");
            }
            // ok, this PDU is to be modified, so can now store it in our list
            v_PDU_SessionId := oct2int(v_GSM_MobilityInfo.SessionId); // @sic R5s200564 sic@
            if (f_IntegerList_Search(v_PDU_SessionIdList, v_PDU_SessionId) < 0) {
              v_PDU_SessionIdList[K] := v_PDU_SessionId;
            } else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "There has been a PDU SESSION MODIFICATION REQUEST with the same session id already");
            }
            for (i := 0; i < N; i := i + 1) {
              if (p_PduSessionsFromLTE[i].PDU_SessionId == oct2int(v_GSM_MobilityInfo.SessionId)) {
     //             f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(p_PduSessionsFromLTE[i].PDU_SessionId, p_PduSessionsFromLTE[i].QFI), p_PduSessionsFromLTE[i].PdnIndex);
                v_GSM_MobilityInfo.EPS_Bearer := valueof(p_PduSessionsFromLTE[i].EPS_Bearer); // @sic R5s210408 sic@
                v_GSM_MobilityInfo.PdnIndex := valueof(p_PduSessionsFromLTE[i].PdnIndex); // @sic R5s210408 sic@
                v_ReregisterIMS := v_ReregisterIMS or p_PduSessionsFromLTE[i].IsIMS;
                break;
              }
            }
            f_NR5GC_MobileInfo_AddPDUSession(v_GSM_MobilityInfo); // @sic R5s201413, R5s210408 sic@
            // At the moment no DRBs are configured in the prose - so this will only work for HO TCs
            if (v_ReregisterIMS) { // @sic R5-205942 sic@
              // Tell IMS to start waiting for the Register message.
              f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_ReregistrationStart(NR));
            }
            v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo);
            f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,
                                             v_CellGroupConfig,
                                             v_RadioBearerList,
                                             v_SDAP_ConfigurationList,
                                             p_NR_CellId,
                                             {v_GSM_MobilityInfo},
                                             v_SRBs);  // @sic R5-213442 sic@
            if (v_SRB == tsc_NR_RbId_SRB1) { // @sic R5s201654, R5s210672 sic@
              v_SRBs := SRB1_2;
              v_SRB := tsc_NR_RbId_SRB2;
            }
            if (p_DRBsAlreadyConfigured) {  // @sic R5-213442 sic@
              SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu))); // @sic R5s201413, R5s210672 sic@
              v_WaitForComplete := true;  // @sic R5-215678 sic@
            } else {                        // @sic R5-213442 sic@
              // Send with a RRC Reconfiguration message
              f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                                 v_RadioBearerConfig,
                                                 v_RadioBearerList,
                                                 v_CellGroupConfig,
                                                 { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu)},
                                                 -, -, false); // @sic R5-215678 sic@
            }
            v_WaitForComplete := true;
            v_WaitForRRCComplete := true; // @sic R5-215678 sic@
            t_Wait.start;  // restart timer to wait for next Request
          }
        }
        else if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
          // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
          if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu, v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
          }
          v_WaitForComplete := false;
          K := K + 1;
        }
        else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS PDU");
        }

      }
      v_RxdMsg := false; // @sic R5s210520 sic@
    }
    // Timer will have been started again in loop, but loop is now finished, so need to stop timer
    t_Wait.stop;
    
    return v_ReregisterIMS;
  }




After Change:
	  
  function f_NR5GC_IRAT_S1ToN1_SessionModification(NR_CellId_Type p_NR_CellId,
                                                   EUTRA_NR_PduSessionInfoList_Type p_PduSessionsFromLTE,
                                                   boolean p_FailWhenK_LessN := true,
                                                   NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                                                   integer p_ExpectedNumberOfNewPDUSessions := (pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection +1),  // @sic R5s220577 sic@
                                                   boolean p_DRBsAlreadyConfigured := false) runs on NR5GC_PTC return boolean
  {
    var template (value) PDUSessionInfoList_Type v_FromLTEList;
    var NG_NAS_UL_PduList_Type v_PduSessionModificationRequestList := {};
    var IntegerList_Type v_PDU_SessionIdList := {};
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_NG_NAS_UL_Pdu;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var integer v_PDU_SessionId;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)
    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;
    var boolean v_ReregisterIMS := false;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var integer N := p_ExpectedNumberOfNewPDUSessions;  // number of PDU SESSION MODIFICATION REQUEST messages being expected
    var integer K := 0;   // number of PDU SESSION MODIFICATION REQUEST messages already processed (i.e. the corresponding PDU SESSION MODIFICATION COMMAND and PDU SESSION MODIFICATION COMPLETE have been sent/received)
    var integer L := 0;   // number of PDU SESSION MODIFICATION REQUEST messages being received
    var boolean v_WaitForComplete := false;
    var template (present) NG_UE_SM_Cap v_NG_UE_SM_Cap := ?;                         // according to 38.508-1 Table 4.5A.2C.2.3-1
    var template (present) IntegrityProtMaxDataRate v_IntegrityProtMaxDataRate := ?; // according to 38.508-1 Table 4.5A.2C.2.3-1
    var integer i;
    var SRBs_AlreadyConfigured v_SRBs := SRB1;
    var NR_RadioBearerId_Type v_SRB := p_SRB;  // @sic R5s210672 sic@
    var IntegerList_Type v_TransferredSessions;
    var PDU_PDN_DNN_Type v_PDU_Type := Internet_DNN; // assume this for now @sic R5-206296 sic@
    var boolean v_RxdMsg := false;
    var boolean v_WaitForRRCComplete := false;
    
    timer t_Wait := 10.0; // @sic R5-204330 sic@

    // Build the GSMMobilityInfo for all received PDUs, but don't store it (some may be established, rather than modified)
    if (ispresent(v_GMM_MobilityInfo.SessionStatus)) { // @sic R5-206290 sic@
      v_TransferredSessions := f_NG_ConvertPDU_SessionStatus(valueof(v_GMM_MobilityInfo.SessionStatus));
      for (i := 0; i < lengthof (p_PduSessionsFromLTE); i := i + 1) {
        if (f_IntegerList_Search(v_TransferredSessions, p_PduSessionsFromLTE[i].PDU_SessionId) != -1) {
          // don't store if not in transferred list
          if (p_PduSessionsFromLTE[i].IsIMS) {  // @sic R5-206296 sic@
            v_PDU_Type := IMS_DNN;
          }
          v_FromLTEList[i] := cs_GSM_MobilityInfo(int2oct(p_PduSessionsFromLTE[i].PDU_SessionId, 1), -, v_PDU_Type); // @sic R5s200564 sic@
        }
      }
    }
        
    t_Wait.start;  // @sic R5-204330 sic@
    while ((K < N) or v_WaitForRRCComplete) { // @sic R5-215678 sic@
      if (not (v_WaitForComplete) and f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L) { // @sic R5s210520, R5s220174, R5s220577 sic@
        // 1st msg was received before entering function
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(L+1); // @sic R5s220174, R5s220577  sic@
        v_RxdMsg := true;
        t_Wait.stop;
      } else {
        alt { // @sic R5-204330 sic@
          [v_WaitForRRCComplete] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB1,
                                      cr_38508_RRCReconfigurationComplete))     // @sic R5s211280 sic@
          {
            v_WaitForRRCComplete := false;
          }
          [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp // @sic R5s201413, R5s210672 sic@
          {
            v_RxdMsg := true;
            t_Wait.stop;
          }
          [] t_Wait.timeout {
            if (p_FailWhenK_LessN and (K < N)) {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not enough PDU Session Modification Request messages");
            } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop  @sic R5-204330 sic@
              break;
            }
          }
        }
      }
      if (v_RxdMsg) { // @sic R5s210520 sic@
        v_NG_NAS_UL_Pdu := v_ReceivedAsp.Signalling.Nas[0].Pdu;
        v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;
 
        if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
             // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
            if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu,     v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
            }
            v_WaitForComplete := false;
            K := K + 1;

            if (L > K)
            {
                //process next PDU SESSION MODIFICATION REQUEST (immediately)
                v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];
                v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;
                L := L-1;
            }
        }

        if (ischosen(v_GSMMsg.pdu_Session_Modification_Request)) {
          v_PduSessionModificationRequestList[L] := v_NG_NAS_UL_Pdu;
          L := L + 1;
          if (L > N) { // @sic R5-204330 sic@
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "L > N");
          }
          if (not v_WaitForComplete and (K < L)) {   // => process the next PDU SESSION MODIFICATION REQUEST
            v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];

            if (not f_Check_NG_PDUSessionModificationReq(v_GSM_MobilityInfo, valueof(v_FromLTEList), v_NG_NAS_UL_Pdu, -, -, v_NG_UE_SM_Cap, -, -, -, v_IntegrityProtMaxDataRate, -, -, -, -, -, cr_NG_Request_Type('101'B))) {     // v_GSM_MobilityInfo is out parameter of f_Check_NG_PDUSessionModificationReq
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION REQUEST");
            }
            // ok, this PDU is to be modified, so can now store it in our list
            v_PDU_SessionId := oct2int(v_GSM_MobilityInfo.SessionId); // @sic R5s200564 sic@
            if (f_IntegerList_Search(v_PDU_SessionIdList, v_PDU_SessionId) < 0) {
              v_PDU_SessionIdList[K] := v_PDU_SessionId;
            } else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "There has been a PDU SESSION MODIFICATION REQUEST with the same session id already");
            }
            for (i := 0; i < N; i := i + 1) {
              if (p_PduSessionsFromLTE[i].PDU_SessionId == oct2int(v_GSM_MobilityInfo.SessionId)) {
     //             f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(p_PduSessionsFromLTE[i].PDU_SessionId, p_PduSessionsFromLTE[i].QFI), p_PduSessionsFromLTE[i].PdnIndex);
                v_GSM_MobilityInfo.EPS_Bearer := valueof(p_PduSessionsFromLTE[i].EPS_Bearer); // @sic R5s210408 sic@
                v_GSM_MobilityInfo.PdnIndex := valueof(p_PduSessionsFromLTE[i].PdnIndex); // @sic R5s210408 sic@
                v_ReregisterIMS := v_ReregisterIMS or p_PduSessionsFromLTE[i].IsIMS;
                break;
              }
            }
            f_NR5GC_MobileInfo_AddPDUSession(v_GSM_MobilityInfo); // @sic R5s201413, R5s210408 sic@
            // At the moment no DRBs are configured in the prose - so this will only work for HO TCs
            if (v_ReregisterIMS) { // @sic R5-205942 sic@
              // Tell IMS to start waiting for the Register message.
              f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_ReregistrationStart(NR));
            }
            v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo);
            f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,
                                             v_CellGroupConfig,
                                             v_RadioBearerList,
                                             v_SDAP_ConfigurationList,
                                             p_NR_CellId,
                                             {v_GSM_MobilityInfo},
                                             v_SRBs);  // @sic R5-213442 sic@
            if (v_SRB == tsc_NR_RbId_SRB1) { // @sic R5s201654, R5s210672 sic@
              v_SRBs := SRB1_2;
              v_SRB := tsc_NR_RbId_SRB2;
            }
            if (p_DRBsAlreadyConfigured) {  // @sic R5-213442 sic@
              SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu))); // @sic R5s201413, R5s210672 sic@
              v_WaitForComplete := true;  // @sic R5-215678 sic@
            } else {                        // @sic R5-213442 sic@
              // Send with a RRC Reconfiguration message
              f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                                 v_RadioBearerConfig,
                                                 v_RadioBearerList,
                                                 v_CellGroupConfig,
                                                 { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu)},
                                                 -, -, false); // @sic R5-215678 sic@
            }
            v_WaitForComplete := true;
            v_WaitForRRCComplete := true; // @sic R5-215678 sic@
            t_Wait.start;  // restart timer to wait for next Request
          }
        }
       /* REMOVED  else if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
          // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
          if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu, v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
          }
          v_WaitForComplete := false;
          K := K + 1;
        }
        else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS PDU");
        }*/

      }
      v_RxdMsg := false; // @sic R5s210520 sic@
    }
    // Timer will have been started again in loop, but loop is now finished, so need to stop timer
    t_Wait.stop;
    
    return v_ReregisterIMS;
  }



 
